restrictions of enumerations

Johan G. F. Belinfante
2010 December 6

In[1]:= **SetDirectory["l:"]; << goedel.10dec05a**

```
:Package Title: goedel.10dec05a                2010 December 5 at 11:00 a.m.

It is now:  2010 Dec 6 at 15:0

Loading Simplification Rules

TOOLS.M is now incorporated in the GOEDEL program as of 2010 September 3

weightlimit = 40
```

summary

The co-restriction of an enumeration to an ordinal is a partial enumeration. For any ordinal α, the co-restriction **id[α] ∘ enum[x]** is the enumeration function for $\mathbf{x} \cap \alpha$.

derivation

Lemma.

In[2]:= **SubstTest[equal, u, composite[funpart[v], id[domain[u]]],
{v → enum[x], u → composite[id[ord[y]], enum[x]]}] // Reverse**

Out[2]= equal[composite[enum[x], id[image[inverse[enum[x]], ord[y]]]],
composite[id[ord[y]], enum[x]]] == True

In[3]:= **composite[enum[x_], id[image[inverse[enum[x_]], ord[y_]]]] :=
composite[id[ord[y]], enum[x]]**

Theorem.

In[4]:= **SubstTest[member, composite[enum[y], id[ord[t]]],
partenum[y], t -> image[inverse[enum[y]], ord[x]]] // Reverse**

Out[4]= member[composite[id[ord[x]], enum[y]], partenum[y]] == True

In[5]:= **member[composite[id[ord[x_]], enum[y_]], partenum[y_]] := True**

Corollary.

```
In[6]:= SubstTest[implies, equal[t, ord[y]],
          member[composite[id[t], enum[x]], partenum[x]], y → t] // Reverse

Out[6]= or[member[composite[id[t], enum[x]], partenum[x]], not[member[t, OMEGA]]] == True

In[7]:= (% /. {t → t_, x → x_}) /. Equal → SetDelayed
```

Theorem. (Result of eliminating the variable **t**.)

```
In[8]:= Map[equal[V, #] &, SubstTest[class, t, not[member[t, w]], w → dif[OMEGA,
            image[inverse[IMAGE[composite[id[enum[x]], inverse[SECOND]]]], partenum[x]]]]]

Out[8]= subclass[image[IMAGE[composite[id[enum[x]], inverse[SECOND]]], OMEGA], partenum[x]] ==
         True

In[9]:= subclass[
          image[IMAGE[composite[id[enum[x_]], inverse[SECOND]]], OMEGA], partenum[x_]] := True
```

Observation. The union of all such co-restrictions of **enum[x]** is the complete enumeration **enum[x]**.

```
In[10]:= U[image[IMAGE[composite[id[enum[x]], inverse[SECOND]]], OMEGA]]

Out[10]= enum[x]
```

Theorem. Each co-restriction to an ordinal belongs to the class of all co-restrictions to ordinals.

```
In[11]:= SubstTest[implies, subclass[u, v], subclass[image[t, u], image[t, v]],
           {t -> IMAGE[composite[id[enum[x]], inverse[SECOND]]],
            u → set[ord[y]], v → OMEGA}] // Reverse

Out[11]= member[composite[id[ord[y]], enum[x]],
           image[IMAGE[composite[id[enum[x]], inverse[SECOND]]], OMEGA]] == True

In[12]:= member[composite[id[ord[y_]], enum[x_]],
           image[IMAGE[composite[id[enum[x_]], inverse[SECOND]]], OMEGA]] := True
```

reverse direction

A similar theorem can be derived that interchanges the roles of restrictions and corestrictions, but offhand it does not appear to be quite as interesting.

Lemma.

```
In[13]:= SubstTest[equal, u, composite[funpart[v], id[domain[u]]],
           {v → inverse[enum[y]], u → composite[id[ord[x]], inverse[enum[y]]]}] // Reverse

Out[13]= equal[composite[id[ord[x]], inverse[enum[y]]],
           composite[inverse[enum[y]], id[image[enum[y], ord[x]]]]] == True

In[14]:= composite[inverse[enum[y_]], id[image[enum[y_], ord[x_]]]] :=
          composite[id[ord[x]], inverse[enum[y]]]
```

Theorem.

```
In[15]:= composite[id[image[enum[x], ord[y]]], enum[x]] // DoubleInverse

Out[15]= composite[id[image[enum[x], ord[y]]], enum[x]] == composite[enum[x], id[ord[y]]]

In[16]:= composite[id[image[enum[x_], ord[y_]]], enum[x_]] := composite[enum[x], id[ord[y]]]
```

enumeration of an intersection with an ordinal

Theorem. Every restriction of **enum[x]** belongs to **ENUMS**.

```
In[17]:= SubstTest[implies, and[member[t, u], subclass[u, v]], member[t, v],
           {t -> composite[enum[x], id[ord[y]]], u → partenum[x], v → ENUMS}] // Reverse

Out[17]= member[composite[enum[x], id[ord[y]]], ENUMS] == True

In[18]:= member[composite[enum[x_], id[ord[y_]]], ENUMS] := True
```

Theorem. Every corestriction of **enum[x]** belongs to **ENUMS**.

```
In[19]:= SubstTest[implies, and[member[u, v], subclass[v, w]], member[u, w],
           {u → composite[id[ord[x]], enum[y]], v → partenum[y], w → ENUMS}] // Reverse

Out[19]= member[composite[id[ord[x]], enum[y]], ENUMS] == True

In[20]:= member[composite[id[ord[x_]], enum[y_]], ENUMS] := True
```

The next theorem is an application of the uniqueness theorem for enumerations. Enumerations with the same range are equal.

Theorem. The co-restriction of **enum[x]** to an ordinal α is the enumeration of $\mathbf{x} \cap \alpha$.

```
In[21]:= SubstTest[implies,
           and[member[u, ENUMS], member[v, ENUMS], equal[range[u], range[v]]], equal[u, v],
           {u -> composite[id[ord[y]], enum[x]], v -> enum[intersection[x, ord[y]]]}] // Reverse

Out[21]= equal[composite[id[ord[y]], enum[x]], enum[intersection[x, ord[y]]]] == True

In[22]:= enum[intersection[x_, ord[y_]]] := composite[id[ord[y]], enum[x]]
```